# $Id: PKGBUILD 85237 2013-02-27 21:14:03Z alucryd $
# Maintainer: Alucryd <alucryd at gmail dot com>

pkgname=faience-themes
pkgver=0.5.4
pkgrel=2
pkgdesc="GTK3, GTK2, Metacity and Gnome-Shell themes based on Faenza"
arch=('any')
url="http://tiheum.deviantart.com/art/Gnome-Shell-Faience-255097456"
license=('GPL3')
optdepends=('gnome-shell: Shell theme'
            'gtk-engine-unico: GTK3 theme'
            'gtk-engine-murrine: GTK2 theme'
            'gnome-tweak-tool: Easy theme switcher'
            'bash: Included resources scripts')
source=("${pkgname}-${pkgver}.zip::http://www.deviantart.com/download/255097456/gnome_shell___faience_by_tiheum-d47vmgg.zip")
sha256sums=('59e7330e01eec62b98aca3b5d9bb6563a22abeed05cfdea2ed96c848d8a81ddb')

package() {
  cd "${srcdir}"

# Install
  mkdir -p "${pkgdir}"/usr/share/themes
  cp -dr --no-preserve=ownership Faience* "${pkgdir}"/usr/share/themes/

# Fix permissions
  find "${pkgdir}" -type d -exec chmod 755 {} +
  find "${pkgdir}" -type f -exec chmod 644 {} +
  chmod 755 "${pkgdir}"/usr/share/themes/Faience{,-Ocre}/gtk-3.0/*.sh
}

# vim: ts=2 sw=2 et:
